

Lumped Equivalent Circuits of Magnetic Components: The Gyrator–Capacitor Approach

David C. Hamill, *Member, IEEE*

Abstract—In place of the conventional reluctance–resistance analogy for forming lumped equivalent circuits of inductive components, a permeance–capacitance analogy is advocated. In this approach, magnetic paths are modeled by capacitive circuits and windings are represented by gyrator two-ports. The technique is applied to the integrated magnetics of a zero-ripple isolated Ćuk dc–dc converter, allowing its electrical and magnetic circuits to be simultaneously simulated with SPICE.

I. INTRODUCTION

CIRCUITS including inductive components with multiple windings are characterized by two types of interconnection: electrical and magnetic. The magnetic connections often give rise to conceptual and analytical difficulties. In an attempt to overcome them, an equivalent circuit model is sought that represents a complex, distributed magnetic circuit by a lumped electrical network with similar behavior. The traditional equivalent circuit is the inductance model, which contains simple inductors and ideal transformers. Unfortunately, the model is not universally valid, and the process of obtaining it hides important relationships among the magnetic circuit variables.

This paper first reviews traditional inductive modeling, which is based on a reluctance–resistance analogy; then the case is argued for an alternative permeance–capacitance analogy. In the resulting equivalent circuit, a capacitive model of the magnetic component is linked to the external electrical circuit by gyrators, which represent windings. To demonstrate its utility in power electronics, gyrator–capacitor modeling is applied to a zero-ripple Ćuk dc–dc converter employing integrated magnetics. Simultaneous simulation of the electrical and magnetic circuits is performed with SPICE. Finally, extension of the technique to lossy and nonlinear materials is outlined.

II. THE TRADITIONAL APPROACH

When considering magnetic circuits from the viewpoint of electronics, it is natural to regard magnetomotive force (mmf) as analogous to voltage and magnetic flux as analogous to current. This traditional pairing results in the reluctance–resistance analogy for modeling magnetic components. The mmf produced by an N -turn winding carrying a current of $i\{A\}$ is $F = Ni\{A\}$. (Where relevant, the SI units of a quantity are appended in braces $\{\cdot\}$. Since N is dimensionless, the units of mmf are properly amperes, but ampere-turns are

TABLE I
TRADITIONAL ANALOGS

Magnetic Circuit		Electrical Circuit	
mmf	$F\{A\}$	Voltage	$v\{V\}$
Flux	$\Phi\{Wb\}$	Current	$i\{A\}$
Reluctance	$\mathcal{R}\{H^{-1}\}$	Resistance	$R\{\Omega\}$
Permeability	$\mu\{H/m\}$	Conductivity	$\sigma\{S/m\}$

in common use.) mmf is visualized as a forcelike quantity that pushes a magnetic flux Φ around the magnetic circuit. The SI unit of flux is the weber $\{Wb\}$ or volt-seconds (per turn) $\{V\cdot s\}$. In a linear, lossless magnetic material, mmf and flux are proportional. Reluctance $\mathcal{R}\{H^{-1}\}$ is then the counterpart of resistance $R\{\Omega\}$ in an electrical circuit: in a magnetic circuit, $\Phi = F/\mathcal{R}$, corresponding directly to Ohm's law, $i = v/R$. Moreover, a sample of magnetic material of permeability $\mu = \mu_0\mu_r$, path length ℓ , and cross-sectional area A has a reluctance of $\mathcal{R} = \ell/\mu A$, a formula that is closely similar to that for electrical resistance, $R = \ell/\sigma A$, where σ is conductivity. The traditional analogs are summarized in Table I.

The resistance model of a magnetic component is found by applying the reluctance–resistance analogy to each section of the magnetic circuit. For example, Fig. 1 shows a leaky two-winding transformer and its resistance model. The equivalent circuit is topologically similar to the magnetic structure; each winding mmf of $F\{A\}$ is represented by a voltage source of $F\{V\}$, and each reluctance of $\mathcal{R} = \ell/\mu A\{H^{-1}\}$ is represented by a resistance of $\mathcal{R}\{\Omega\}$. The currents $\{A\}$ in the electrical circuit correspond to fluxes $\{Wb\}$ in the magnetic circuit. The inductance of a winding may be determined by finding the effective reluctance \mathcal{R}_{eff} seen by the winding mmf source; if the winding consists of N turns, its inductance is $L = N^2/\mathcal{R}_{eff}$.

If the winding currents are known, the fluxes can be determined in a straightforward manner. However, in many cases of practical interest, the magnetic and electrical circuits interact, and the system has to be analyzed as a whole. To allow this, the resistance model is converted into an inductance model, which consists of a network of simple inductors and ideal transformers [1], [2]. Unlike the resistance model, the inductance model can be connected directly into the surrounding electrical circuit. The conversion procedure comprises two steps.

1) The dual of the resistance model is formed by applying the laws of electrical voltage–current duality. Network loops

Manuscript received November 26, 1991; revised October 28, 1992.

The author is with the Department of Electronic and Electrical Engineering, University of Surrey, Guildford GU2 5XH, England.

IEEE Log Number 9206669.




Fig. 1. Magnetic circuit of a leaky two-winding transformer and its resistance model.




Fig. 2. Conversion of the resistance model of Fig. 1 into an inductance model.

become nodes, and vice versa; each voltage source $F\{V\}$ becomes a current source of $F\{A\}$, and each resistance of $R\{\Omega\}$ is replaced by a conductance of $R\{S\}$.

2) A particular winding is chosen as the reference winding, having N_r turns, say. The associated current source is replaced by a pair of terminals. The current source associated with every other winding of N_i turns is replaced by an ideal transformer of ratio $N_r : N_i$ feeding a pair of terminals. Each conductance of $R_j\{S\}$ is replaced by an inductance of $N_r^2 R_j\{H\}$.

Fig. 2 illustrates the two steps for the resistance model of Fig. 1.

Although this procedure is mechanical, it is not universally successful: if the magnetic circuit is nonplanar, the procedure fails because, as proved by elementary graph theory, nonplanar networks have no dual. A nonplanar network is one whose schematic cannot be drawn without crossovers. More exactly, a network is nonplanar if its graph contains either of the subgraphs of Fig. 3 [3]. In general, magnetic components with more than three windings are nonplanar if all possible couplings are taken into account [2]. Fig. 4 shows an example that could be constructed from three standard E-cores; because it is topologically nonplanar, no simple inductance model is possible for this component.¹

A recently introduced technique [5], [6] retains the resistance model, linking it to the external electrical circuit via a "magnetic interface." The interface implements the pair of equations governing a winding:

$$\begin{aligned} v &= N \frac{d\Phi}{dt} \\ F &= Ni. \end{aligned} \quad (1)$$

¹The restriction can be overcome by employing additional ideal transformers [4], although the resulting inductance model may perhaps be difficult to understand.




Fig. 3. Subgraphs of nonplanar networks.




Fig. 4. Example of a nonplanar magnetic component.

A six-line SPICE netlist of the interface circuit comprises three controlled sources, two independent sources, and an inductor that performs the differentiation. Simultaneous simulation of the electrical and magnetic circuits is possible for both planar and nonplanar magnetics.

Although it has served for many years, confidence in the traditional resistance model is undermined by a simple question: magnetic reluctances *store* energy, so why are they made analogous to electrical resistances, which *dissipate* energy? The objection is particularly worrying in power electronics, where energy relations are of prime importance. At the root of the problem is the initial choice of mmf and flux as the "natural" magnetic circuit variables.

A generalized energy-based network, such as an electrical, hydraulic, or mechanical system, is characterized by *effort* variables and *flow* variables [7] (also known as *across* and *through* variables, respectively). The system variables are usually chosen so that when an effort and its corresponding flow are multiplied together, the result has the dimensions of power. The examples of Table II illustrate the point. In the familiar electrical case, voltage is the effort variable that pushes charge around the circuit, giving rise to the flow variable, current. The product of voltage and current is power, as confirmed by a check on their units:

$$\begin{aligned} \{V\} &= \{J/C\}; \quad \{A\} = \{C/s\}; \\ \{V\} \cdot \{A\} &= \{J/s\} = \{W\}. \end{aligned} \quad (2)$$

In hydraulic and mechanical systems, the product of effort and flow variables is again power. However, with the conventional choice of mmf and flux as the magnetic system variables, the product of effort and flow is energy:

$$\begin{aligned} \{A\} &= \{C/s\}; \quad \{W\} = \{V \cdot s\} = \{J \cdot s/C\}; \\ \{A\} \cdot \{Wb\} &= \{J\}. \end{aligned} \quad (3)$$

TABLE II
EFFORT AND FLOW VARIABLES IN VARIOUS DOMAINS

Domain	Effort Variable	Flow Variable	Effort · Flow Product
Electrical	Voltage {V}	Current {A}	Power {V · A} ≡ {W}
Hydraulic	Pressure {N/m ² }	Volume flow {m ³ /s}	Power {N · m/s} ≡ {W}
Rectilinear mechanical	Force {N}	Velocity {m/s}	Power {N · m/s} ≡ {W}
Rotating mechanical	Torque {N · m}	Angular velocity {rad/s} ≡ {s ⁻¹ }	Power {N · m/s} ≡ {W}
Magnetic (traditional)	mmf {A}	Magnetic flux {Wb} ≡ {V · s}	Energy {V · A · s} ≡ {J}
Magnetic (alternative)	mmf {A}	Magnetic flux rate {Wb/s} ≡ {V}	Power {V · A} ≡ {W}

TABLE III
ALTERNATIVE ANALOGS

Magnetic Circuit			Electrical Circuit		
mmf	F	{A}	Voltage	v	{V}
Flux rate	$\dot{\Phi}$	{V}	Current	i	{A}
Permeance	\mathcal{P}	{H}	Capacitance	C	{F}
Flux	$\Phi = \int \dot{\Phi} dt$	{Wb}	Charge	$q = \int i dt$	{C}
Permeability	$\mu = \mu_0 \mu_r$	{H/m}	Permittivity	$\epsilon = \epsilon_0 \epsilon_r$	{F/m}
Power	$P = F \dot{\Phi}$	{W}	Power	$P = vi$	{W}
Energy	$E = \int F d\Phi$	{J}	Energy	$E = \int v dq$	{J}

For consistency with an electrical equivalent circuit, the product should be power, not its integral. The energy stored in a magnetic component should equal that stored in its equivalent circuit, but there is no energy stored in the resistance model. Thus, for understanding energy relations and dynamics in the context of power electronics, the conventional resistance-reluctance analogy can lead to much confusion.

III. THE CAPACITANCE MODEL

In the late 1960's, Buntenbach proposed an alternative analogy [8]–[10] in which mmf is retained as the magnetic effort variable, but the rate-of-change of flux ($d\Phi/dt \equiv \dot{\Phi}$) is chosen as the flow variable. It is convenient to call $\dot{\Phi}$ the *flux rate*; its units are webers/second or volts (per turn) {V}. The product of effort and flow variables is then power. In constructing an electrical equivalent circuit, an analogy is drawn between mmf and voltage as effort variables and flux rate and current as flow variables. In the new system, magnetic flux is analogous to electric charge, not electric current. Just as voltage pushes charge around the electrical circuit causing a flow of current ($i = dq/dt$), so mmf pushes flux around the magnetic circuit, causing a flow of flux rate ($\dot{\Phi} = d\Phi/dt$).

With the new variables, magnetic permeance becomes analogous to electrical capacitance. This may be seen as follows: $\Phi = F/\mathcal{R} = \mathcal{P}F$, where \mathcal{P} {H} = $1/\mathcal{R}$ is permeance; hence, $\dot{\Phi} = \mathcal{P} dF/dt$. This differential equation corresponds to that governing capacitance in an electrical circuit, $i =$

$C dv/dt$. The formula for calculating permeance, $\mathcal{P} = \mu A/\ell$, also corresponds to that for capacitance, $C = \epsilon A/\ell$, where ϵ is permittivity. A magnetic structure may be represented by a topologically similar capacitance model, each lumped permeance of \mathcal{P} {H} corresponding to a capacitance of \mathcal{P} {F}. In this respect, the capacitance model follows the resistance model, but with capacitors in place of resistors; however, the quantity and distribution of energy within the magnetic circuit are now correctly represented in the model. The alternative analogs are summarized in Table III.

IV. WINDINGS AS GYRATORS

A distinctive feature of Buntenbach's approach is the way in which windings are treated. A winding may be thought of as a two-port element that links the electrical and magnetic circuits. An N -turn winding relates variables v and i at the electrical port to variables F and $\dot{\Phi}$ at the magnetic port:

$$\begin{aligned} v &= N \dot{\Phi} \\ i &= F/N. \end{aligned} \quad (4)$$

Thus, the electrical effort variable v is proportional to the magnetic flow variable $\dot{\Phi}$, while the electrical flow variable i is proportional to the magnetic effort variable F . A relationship where effort and flow are exchanged is characteristic of a *gyrator*, an ideal, lossless, two-port electrical circuit element [7], [11]. With voltage and currents defined as in Fig. 5, it is




Fig. 5. The gyrator two-port.




Fig. 6. Reference directions of electrical and magnetic circuit variables.

governed by the equations

$$\begin{aligned} v_1 &= r i_2 \\ i_1 &= v_2 / r. \end{aligned} \quad (5)$$

(The reference direction for i_2 adopted here is the opposite of the usual two-port convention but is more convenient where energy transfer is of interest.) The quantity $r \{\Omega\}$ is termed the gyrator modulus, or gyration resistance. The gyrator is a more fundamental circuit element than the ideal transformer: an ideal transformer can be produced by cascading two ideal gyrators, but a gyrator cannot be produced from transformers [11].

An appropriate alias for the gyrator might be the "dualizer": by interchanging the roles of voltage and current, a gyrator turns an impedance into its dual. When an impedance Z is connected to one of its ports, an impedance of r^2/Z is seen at the other; for example, when a $1 \mu\text{F}$ capacitor, impedance $Z = 10^6/j\omega$, is connected to a gyrator of modulus 10Ω , it appears at the other port as an impedance $10^2/Z = 10^{-4} j\omega$, i.e., an inductance of $100 \mu\text{H}$. Comparing the winding and gyrator equations, (4) and (5), it is seen that an N -turn winding acts as an $N\text{-}\Omega$ gyrator. (Although N is physically dimensionless, when electrical variables are substituted for the magnetic ones, a change of units occurs, and N must be in ohms for dimensional consistency.) The inductance Z of a winding may be determined by finding the effective permeance \mathcal{P}_{eff} loading the corresponding gyrator; the inductance is then calculated from $L = N^2 \mathcal{P}_{\text{eff}}$. The notion of an effective permeance will already be familiar to magnetics designers as the inductance factor A_L , used to characterize core assemblies.

Fig. 6 sets out the reference directions for the electrical and magnetic variables associated with a winding and its equivalent gyrator. These reference directions are compatible with the conventional right-hand rule: with the right thumb pointing in the reference direction of i , the curved fingers point in the reference direction of Φ .




Fig. 7. Gapped-core inductor and its gyrator-capacitor model.

Ideal gyrators are easily implemented in a circuit simulator such as SPICE in two different ways, each of which involves a pair of controlled sources. The first method employs a pair of current-controlled voltage sources (CCVS). The gyrator equations (5) may be written as $v_1 = r i_2, v_2 = r i_1$ and implemented in SPICE, e.g., for $r = 10 \Omega$, as

```
V01 1 5
V02 6 3
H1 5 2 V02 10
H2 6 4 V01 10.
```

Zero-voltage sources $V01$ and $V02$ are used as current sensors because standard SPICE does not allow the current through a CCVS to be sensed directly. The second method uses a pair of voltage-controlled current sources (VCCS). The gyrator equations are recast as $i_1 = g v_2, i_2 = g v_1$, where $g = 1/r$. The modified equations are implemented in SPICE, e.g., for $g = 0.1 \text{ S}$, as

```
G1 1 2 3 4 0.1
G2 4 3 1 2 0.1.
```

Although slightly more complex, the CCVS version is usually preferable for magnetic modeling purposes for two reasons. First, it allows the series connection of mmf's, corresponding to tightly coupled windings, whereas series-connected VCCS's would suffer from current contention problems. Second, N appears directly as an integer in the listing of the CCVS version rather than as $1/N$, which is less convenient and may be prone to entry errors.

V. EXAMPLES

The gyrator-capacitor approach is illustrated next by two simple examples. In each case, the equivalent circuit is formed from the magnetic circuit in a single step.

1) *Gapped-Core Inductor*: The core in Fig. 7 has permeability $\mu_0 \mu_r$, cross-sectional area A , path length ℓ_c , and gap length ℓ_g , and is wound with N turns. Therefore, the permeance of the core is $\mathcal{P}_c = \mu_0 \mu_r A / \ell_c$ and that of the airgap is $\mathcal{P}_g = \mu_0 A / \ell_g$, neglecting fringing. In the gyrator-capacitor model of the assembly, the effective permeance loading the gyrator corresponds to the series combination of two capacitors, so the inductance seen by the external circuit is $L = N^2 / (1/\mathcal{P}_c + 1/\mathcal{P}_g)$.

2) *Two-Winding Transformer*: Fig. 8 shows a gyrator-capacitor model of the leaky transformer treated in Figs. 1 and 2. If required, the open-circuit and short-circuit inductances of a winding may be found by a method similar to that of the previous example. For instance, if the N_2 winding is





Fig. 8. Gyrator-capacitor model of the transformer of Fig. 1.

Fig. 9. Ćuk converter with integrated magnetics; the output is nominally 5 V, 10 A. S_1 operates at 500 kHz with a duty factor of 0.3. $V_I = 28$ V, $C_1 = 5$ μ F, $C_2 = 20$ μ F, $C_3 = 10$ μ F, $R_L = 0.5$ Ω .

short circuited, the N_1 winding has an inductance of $L_1 = N_1^2 / (1/P_1 + 1/P_5 + 1/P_4)$. (A gyrator mirrors a short circuit as an open circuit, and vice versa.) In the limiting case when P_1 to P_4 become infinite and the leakage permeance P_5 goes to zero, the circuit reduces to a pair of cascaded gyrators, i.e., an ideal transformer of ratio $N_1 : N_2$.

The interested reader may care to sketch an equivalent circuit of the nonplanar component of Fig. 4.

VI. MODELING AND SIMULATION OF A ĆUK CONVERTER

As an illustration of the technique in a practical system with electrical and magnetic parts, it is now applied to a complex dc-dc converter: an isolated Ćuk converter with integrated magnetics, capable of zero-ripple operation. Certain dc-dc converters may be miniaturized by combining several magnetic components into a single assembly. In the isolated Ćuk converter, two chokes and a transformer may be integrated into a single structure. Moreover, given the proper magnetic coupling between windings (the *zero-ripple condition*), the choke ripple currents can be nulled out, resulting in smooth dc at the converter's input and output terminals. Despite several published studies, the zero-ripple mechanism remains obscure to many engineers; this poor understanding seems to be related to the interaction of the magnetic and electrical circuits. Simulation may help to clarify matters, given a suitable model of the integrated magnetics. With this aim in mind, a gyrator-capacitor model is developed and then combined with the electrical circuit, allowing SPICE simulation of the complete converter.

In an entertaining paper [12], Ćuk and co-workers outlined the design of the converter shown in Fig. 9. The core consists of a pair of half-thickness EE-22 cores of H7C4 ferrite (TDK), gapped by a 0.3 mm spacer. The permeances can be calculated approximately by partitioning the magnetic paths into a number of lumped elements. There are many ways to do




Fig. 10. Core and gap geometry used to calculate permeances. All dimensions are in millimeters.

TABLE IV
CORE AND GAP PERMEANCES CALCULATED FROM FIG 10

	A {mm ² }	ℓ {mm}	μ_r	\mathcal{P} {nH}
\mathcal{P}_{c1}	4.0×3.0	7.5	2000	4020
\mathcal{P}_{c2}	4.0×3.0	7.3	2000	4130
\mathcal{P}_{c3}	6.0×3.0	7.3	2000	6200
\mathcal{P}_{g1}	4.3×3.3	0.3	1	59.4
\mathcal{P}_{g2}	6.3×3.3	0.3	1	87.1




Fig. 11. Simplified gyrator-capacitor model of the Ćuk converter's integrated magnetics.

this, but for present purposes, the scheme shown in Fig. 10 is satisfactory, and has the virtue of simplicity. The permeances, estimated from the core geometry, are listed in Table IV. The gap permeances \mathcal{P}_{g1} and \mathcal{P}_{g2} allow for fringing by the usual expedient of increasing each core dimension by the gap length when calculating A . Because some of the core permeances appear in series, they may be combined:

$$\mathcal{P}_1 = \mathcal{P}_3 = \frac{1}{2/\mathcal{P}_{c1} + 2/\mathcal{P}_{c2} + 1/\mathcal{P}_{g1}} = 56.1 \text{ nH} \quad (6)$$

$$\mathcal{P}_2 = \frac{1}{2/\mathcal{P}_{c3} + 1/\mathcal{P}_{g2}} = 84.7 \text{ nH}. \quad (7)$$

As expected, the gap permeances dominate. Replacing \mathcal{P}_1 to \mathcal{P}_3 by capacitors and adding winding gyrators of 16, 6, 3, and 8 n, with due regard to their phasing, the gyrator-capacitor equivalent circuit of Fig. 11 is obtained.

One permeance that is not obvious from the core geometry, yet is important in the operation of the converter, is the leakage path for center-limb flux that bypasses the outer limbs, represented in Fig. 11 by \mathcal{P}_4 . Because the path is through air, \mathcal{P}_4 is difficult to calculate. However, in this instance its value can be deduced because the converter is known to satisfy the zero-ripple condition: the transformer-to-choke coupling coefficient must equal the transformer-to-choke turns ratio (for both the input and output circuits). The coupling coefficient is




Fig. 12. Start-up waveforms of the Ćuk converter, simulated with PSpice. Top to bottom: voltage across R_L ; voltage across S_1 ; current in input-choke winding; current in output-choke winding; flux densities $\{T\}$ in LH limb, center limb, and RH limb of core.

the proportion of center-limb flux rate that links each outer limb. In the equivalent circuit of Fig. 11, the $6\ \Omega$ gyrator produces a flux rate that flows through \mathcal{P}_2 . With the other windings open circuit, the flux rate divides among \mathcal{P}_1 , \mathcal{P}_3 , and \mathcal{P}_4 . Elementary circuit theory allows the proportion flowing through \mathcal{P}_1 to be found, giving the coupling coefficient as $k = \mathcal{P}_1/(\mathcal{P}_1 + \mathcal{P}_3 + \mathcal{P}_4)$. Equating k to the turns ratio of 0.375 (for zero ripple), capacitance \mathcal{P}_4 is calculated as 36.7 nF. If \mathcal{P}_4 were neglected, k would be 0.5 and the zero-ripple condition would be violated.

In the PSpice² simulation, S_1 was a *VSWITCH* device that could be replaced by a transistor in other versions of SPICE. A series $50\ \mu\text{F} + 10\ \Omega$ damping network was added across C_1 to speed convergence to the steady state. A $1\ \text{G}\Omega$ resistance was added between ground and the all-capacitor node to prevent it from floating. (The PSpice netlist is available from the author.) Simulated waveforms of the start-up transient are shown in Fig. 12, which covers 100 switching cycles. Very little switching-frequency ripple current is visible in the input and output choke windings, confirming zero-ripple operation.

The flux in any permeance may be observed by integrating the corresponding capacitor current using the Probe facility of PSpice. For instance, to see the flux in the center leg \mathcal{P}_2 ,

²PSpice® and Probe™ are trademarks of MicroSim Corporation, Irvine, CA.

the current $I(CP2)$ is integrated. The flux $\{Wb\}$ is then numerically equal to the charge $\{C\}$. To find the flux density, the flux value is divided by the appropriate cross-sectional area. The simulated flux-density excursions approach 400 mT, indicating that transient core saturation at start-up may be a problem in the published converter design. As with any integration, the proper initial conditions must be set; here, the capacitor voltages are initially zero, as there is assumed to be no flux prior to start-up. This remark also applies to the magnetic interface of [5], [6], where the integration is performed at run time, rather than by postprocessing.

VII. EXTENSION TO NONIDEAL CASES

Real magnetic materials suffer eddy-current losses due to their finite electrical resistivity. Buntentbach showed that conductive magnetic materials are properly modeled under the permeance-capacitance analogy by a distributed RC network [8]. The saturation and hysteresis properties of real magnetic materials at high field excursions may also be included: to model a saturating $\Phi(F)$ characteristic, the capacitors representing core permeances are given a nonlinear $q(v)$ characteristic, while hysteresis may be modeled, to a useful approximation, by placing a nonlinear resistance in series with the capacitor to give a sigmoid $\Phi(F)$ loop. It is planned to publish further details in a future paper.

VIII. CONCLUSION

Buntentbach's original work on gyrator-capacitor modeling seems to have been largely overlooked. This is a pity because it can bring considerable advantages to the understanding and simulation of magnetic components, for long a problem area for power electronics engineers. The gyrator-capacitor approach is conceptually straightforward and easy to apply because the equivalent circuit is topologically similar to the magnetic component it models. The method can handle both planar and nonplanar magnetic structures, and may be extended to encompass lossy and nonlinear magnetic materials. Unlike traditional modeling, magnetic energy relations are preserved in the equivalent circuit. Gyrator-capacitor models are easily incorporated in simulations, proving particularly valuable for complex electrical-magnetic systems such as power converters with integrated magnetic components. With all of these advantages, the gyrator-capacitor approach deserves to be more widely adopted in power electronics.

REFERENCES

- [1] G. W. Ludwig and S. A. El-Hamamsay, "Coupled inductance and reluctance models of magnetic components," *IEEE Trans. Power Electron.*, vol. 6, pp. 240–250, Apr. 1991.
- [2] E. Colin Cherry, "The duality between interlinked electric and magnetic circuits and the formation of transformer equivalent circuits," *Proc. Phys. Soc. London*, vol. 62B, pp. 101–111, Feb. 1949.
- [3] C. Kuratowski, "Sur le problème des courbes gauches en topologie" (in French), *Fundamenta Mathematicae*, vol. 1, no. 5, pp. 271–283, 1930.
- [4] A. Bloch, "On methods for the construction of networks dual to non-planar networks," *Proc. Phys. Soc.*, vol. 58, pp. 677–694, Nov. 1946.
- [5] S. A. El-Hamamsay and E. I. Chang, "Magnetics modeling for computer-aided design of power electronics circuits," in *Power Electron. Specialists Conf. Rec.*, vol. 2, Milwaukee, WI, June 1989, pp. 635–645.

- [6] M. Kahmann, "Gemischte Simulation elektrischer und magnetischer Netzwerke mit 'Spice' am Beispiel eines Stromwandlers mit M-Kern," *ETZ Archiv* (in German), vol. 12, pp. 21–25, Jan. 1990.
- [7] D. Karnopp and R. Rosenberg, *System Dynamics: A Unified Approach*. New York: Wiley, 1975.
- [8] R. W. Buntentbach, "Improved circuit models for inductors wound on dissipative magnetic cores," in *Proc. 2nd Asilomar Conf. Circuits Syst.*, Pacific Grove, CA, Oct. 1968, pp. 229–236 (IEEE Publ. No. 68C64-ASIL).
- [9] ———, "Analogues between magnetic and electrical circuits," *Electron. Products*, vol. 12, pp. 108–113, Oct. 1969.
- [10] ———, "A generalized circuit model for multiwinding inductive devices" (digest only), *IEEE Trans. Magn.*, vol. MAG-6, p. 65, Mar. 1970.
- [11] B. D. H. Tellegen, "The gyrator, A new electric network element," *Philips Res. Rep.*, vol. 3, pp. 81–101, Apr. 1948.
- [12] S. Cuk, Z. Zhang, and L. A. Kajouke, "Low profile, 50W/in³, 500kHz integrated-magnetics PWM Ćuk converter," in *Proc. High Freq. Power Conversion Conf.*, San Diego, CA, May 1988, pp. 442–463.



David C. Hamill (M'89) was born in London, England. He received the B.Sc.(Eng.) degree in electronic engineering and the M.Sc. degree in sound and vibration from the University of Southampton, Southampton, England, in 1972 and 1974, respectively.

After practicing as an Audio Design Engineer and an Electronics Consultant, he was Technical Director of PAG Ltd. for seven years. In 1986 he joined the Department of Electronic and Electrical Engineering, University of Surrey, as a Lecturer in power electronics. His current research interests include high-frequency dc–dc conversion, chaotic systems, circuit modeling and simulation, and other applications of computers in power electronics. He is at present pursuing the doctoral degree.

Mr. Hamill is a Chartered Engineer and a member of the Institution of Electrical Engineers, UK, and the United Kingdom Simulation Society.

PSpice LISTING

This netlist was used to produce the waveforms of Fig. 12 of the paper "Lumped equivalent circuits of magnetic components: the gyrator-capacitor approach" by D.C. Hamill.

Cuk Converter with Integrated Magnetics

* Integrated magnetics:

X1 1 2 3 4 5 6 INTMAG

* Input section:

VI 1 0 DC 28

S1 2 0 10 0 SWITCH

.MODEL SWITCH VSWITCH

+ (VON=1 VOFF=0 RON=0.1 ROFF=100K)

C1 3 0 5U

RDAMP 3 100 10.0

CDAMP 100 0 50U

* Output section:

C2 4 0 20U

D1 0 5 DIODE

.MODEL DIODE D

C3 6 0 10U

RL 6 0 0.5

* Drive for switch, fs=500kHz, D=0.3:

VDRIVE 10 0 PULSE(0 1 0 50N 50N 600N 2U)

* Transient analysis:

.OPTIONS ITL4=30 ITL5=0 RELTOL=0.01

.TRAN 200N 200U 0 200N

.PROBE

.SUBCKT INTMAG 1 2 3 4 5 6

* Integrated magnetics subcircuit

* External nodes:

* 1-2: input choke winding

* 2-3: primary transformer winding

* 4-5: secondary transformer winding

* 5-6: output choke winding

* Permeances P1 through P4:

CP1 11 13 56.1N

CP2 14 13 84.7N

CP3 19 13 56.1N

CP4 13 0 36.7N

* To prevent node 13 floating:

RFIX 13 0 1G

* Input choke winding

* 16 turns, nodes 1-2:

VS11 1 9

VS12 10 11

H11 9 2 VS12 16.0

H12 10 0 VS11 16.0


```
* Primary winding
* 6 turns, nodes 2-3:
VS21 2 12
VS22 15 14
H21 12 3 VS22 6.0
H22 15 16 VS21 6.0

* Secondary winding
* 3 turns, nodes 4-5:
VS31 16 17
VS32 18 5
H31 17 0 VS32 3.0
H32 18 4 VS31 3.0

* Output choke winding
* 8 turns, nodes 5-6:
VS41 19 20
VS42 21 6
H41 20 0 VS42 8.0
H42 21 5 VS41 8.0

.ENDS INTMAG

.END
```